Coinductive Programming and Proving in Agda

Lecture 1: Coinductive programming in Agda

Jesper Cockx

21 January 2026

Technical University Delft

Lecture plan

1. Introduction
2. Data types and (co)pattern matching
3. Universes and polymorphism
4. Dependent types
5. Coinductive record types
6. Mixing induction and coinduction
7. Sized types

Outline

1 Introduction

2 Data types and (co)pattern matching

3 Universes and polymorphism

4 Dependent types

5 Coinductive record types

6 Mixing induction and coinduction

7 Sized types

Why dependent types?

Dependent types embed formal verification directly inside your type system.

Advantages of dependent types.

- Single syntax for programming and proving
- Editor support for interactive development
- Invariants can be embedded inside programs (= *intrinsic* verification)

Verifying a program should not be more difficult than writing it in the first place!

The Agda language

Agda is a purely functional programming language similar to Haskell.

Unlike Haskell, it has full support for dependent types.

It also supports interactive programming with help from the type checker.

Installing Agda

1. **Agda binary.** Download the binary[1], then run `agda --setup`.
2. **Editor plugin.** Install the VS Code plugin or run `agda --emacs-mode setup`.
3. **Standard library.** Download and unpack[2], then add it to `libraries` and `defaults`.

Detailed instructions:
agda.readthedocs.io/en/v2.8.0/getting-started

[1] github.com/agda/agda/releases/tag/v2.8.0
[2] github.com/agda/agda-stdlib/archive/v2.3.tar.gz

Basic syntax

Names can be any non-reserved sequence of unicode[3] characters, except `. ; { } ( ) "`.

A **type declaration** is written as $b : A$.

Function types are written $A \to B$ or $(x : A) \to B$. $\lambda\, x \to u$ is an (anonymous) function and $f\ x$ is function application.

An **infix operator** `_+_` is used as $x + y$.
`x+y` is a valid name, so use enough spaces.

[3] Supported editors will replace LaTeX-like syntax (e.g. `\to`) with unicode.

Loading an Agda file

You can load an Agda file by pressing `Ctrl+c` followed by `Ctrl+l`.

Once the file is loaded (and there are no errors), other commands become available:

`Ctrl+c Ctrl+d` Infer type of an expression.

`Ctrl+c Ctrl+n` Evaluate an expression.

Holes in programs

A hole is an incomplete part of a program. You can create one by writing `?` or `{!!}` and loading the file.

New commands for holes:

`Ctrl+c Ctrl+,`	Get hole information
`Ctrl+c Ctrl+c`	Case split on a variable
`Ctrl+c Ctrl+space`	Fill in the hole

Outline

1 Introduction

2 Data types and (co)pattern matching

3 Universes and polymorphism

4 Dependent types

5 Coinductive record types

6 Mixing induction and coinduction

7 Sized types

Declaring new datatypes

```
data Bool : Set where
  true  : Bool
  false : Bool

data ℕ : Set where
  zero : ℕ
  suc  : ℕ → ℕ
{-# BUILTIN NATURAL ℕ #-}
```

Set is the type of (small) types (see later).

Defining functions by pattern matching

```
not : Bool → Bool
not true  = false
not false = true

_+_ : ℕ → ℕ → ℕ
zero    + y = y
(suc x) + y = suc (x + y)
```

Exercise. Define isEven, _*_, and _≤_ on ℕ.

Total functional programming

Agda is a total language: evaluating a function call always returns a result in finite time:

- The **coverage checker** ensures completeness of pattern matches.
- The **termination checker** ensures termination of recursive definitions.

Reasons to care about totality:

- It prevents crashes and infinite loops.
- It is needed for **decidable type checking**.
- It is needed for **logical soundness**.

Record types and projections

```
record Rect : Set where
  constructor rect -- optional
  field
    height : ℕ
    width : ℕ

square : ℕ → Rect
square x = record { height = x ; width = x }
    -- or: rect x x

area : Rect → ℕ
area r = Rect.height r + Rect.width r
```

More record syntax

You can open the record to bring projections into scope:

```
open Rect

perimeter : Rect → ℕ
perimeter r = 2 * height r + 2 * width r
```

You can also use projections as postfix:

```
rotate : Rect → Rect
rotate r = rect (r .width) (r .height)
```

Copattern matching

Where data types are defined by how we construct an element, record types are defined by what we can observe of an element.

This duality is exploited by Agda's copattern syntax:

```
squeeze : Rect → Rect
squeeze r .height = half (r .height)
squeeze r .width  = 2 * r .width
```

We define squeeze *r* by its projections.

Outline

1 Introduction

2 Data types and (co)pattern matching

3 Universes and polymorphism

4 Dependent types

5 Coinductive record types

6 Mixing induction and coinduction

7 Sized types

The type `Set`

In Agda, types such as $\mathbb{N}$ and $\mathsf{Bool} \to \mathsf{Bool}$ are themselves expressions of type Set.

We can pass around and return values of type Set just like values of any other type.

Example.

$$\mathsf{id} : (A : \mathsf{Set}) \to A \to A$$
$$\mathsf{id}\ A\ x = x$$

A type like Set whose elements are themselves types is called a universe.

Side note: the `Set` hierarchy

Set itself does not have type Set: assuming so leads to inconsistency.[4]

Instead, Set = Set_0 has type Set_1, which has type Set_2, which has type...

In fact, you can write universe-polymorphic definitions by quantifying over l : Level and working with universe Set l.

[4] See Girard's paradox and Hurken's paradox.

Polymorphic functions in Agda

We can use Set to define polymorphic functions and data types:

```
data List (A : Set) : Set where
  []  : List A
  _::_ : A → List A → List A

length : {A : Set} → List A → ℕ
length []       = zero
length (_ :: xs) = suc (length xs)
```

The curly braces mark A as implicit.

Exercise. Implement map and _++_.

Variable generalization

We can mark declare a variable to be generalized automatically:

```
variable A B C : Set

id : A → A
id x = x
```

This is equivalent to writing $\{A : \mathsf{Set}\} \to \ldots$

Polymorphic record types

```
record _×_ (A B : Set) : Set where
  constructor _,_
  field
    proj₁ : A
    proj₂ : B
open _×_ public

swap : A × B → B × A
swap (x , y) = y , x
```

If/then/else as a function

We can define if/then/else in Agda as follows:

```
if_then_else_ : Bool → A → A → A
if true  then x else y = x
if false then x else y = y
```

This is an example of a mixfix operator.

Example usage.

```
test : ℕ → ℕ
test x = if (x ≤ 9000) then 0 else 42
```

The empty type and absurd patterns

```
data ⊥ : Set where

absurd : ⊥ → A
absurd ()
```

The absurd pattern `()` indicates that there are no possible constructors.

The disjoint sum type

The Either type from Haskell is called $\_\uplus\_$:

```
data _⊎_ (A B : Set) : Set where
  inj₁ : A → A ⊎ B
  inj₂ : B → A ⊎ B

[_,_] : (A → C) → (B → C) → A ⊎ B → C
[ f , g ] (inj₁ x) = f x
[ f , g ] (inj₂ y) = g y
```

Outline

1 Introduction

2 Data types and (co)pattern matching

3 Universes and polymorphism

4 Dependent types

5 Coinductive record types

6 Mixing induction and coinduction

7 Sized types

Dependent types

A dependent type is a family of types, depending on a term of a base type:

```
data Flavour : Set where
  cheesy chocolatey : Flavour

data Food : Flavour → Set where
  pizza  : Food cheesy
  cake   : Food chocolatey
  bread : {f : Flavour} → Food f

amountOfCheese : Food cheesy → ℕ
amountOfCheese pizza  = 100
amountOfCheese bread = 20
```

Agda knows that cake is not a valid input!

Vectors: lists that know their length

Vec A n is the type of vectors of length n:

```
data Vec (A : Set) : ℕ → Set where
  []  : Vec A 0
  _::_ : A → Vec A n → Vec A (suc n)

myVec : Vec ℕ 4
myVec = 1 :: 2 :: 3 :: 4 :: []
```

Types will be normalized during type checking:

```
myVec' : Vec ℕ (2 + 2)
myVec' = myVec
```

Side note: Parameters vs. indices

The argument $(A : Set)$ in the definition of Vec is a parameter, and has to be *the same in the type of each constructor*.

The argument of type $\mathbb{N}$ in the definition of Vec is an index, and must be *determined individually for each constructor*.

Quiz question

Question. How many elements are there in the type Vec Bool 3?

Quiz question

Question. How many elements are there in the type `Vec Bool 3`?

Answer. 8 elements:

- `true :: true :: true :: []`
- `true :: true :: false :: []`
- `true :: false :: true :: []`
- `true :: false :: false :: []`
- `false :: true :: true :: []`
- `false :: true :: false :: []`
- `false :: false :: true :: []`
- `false :: false :: false :: []`

Dependent function types

A dependent function type is a type of the form $(x : A) \to B\ x$ where the *type* of the output depends on the *value* of the input.

Example.

$$\begin{aligned}
&\mathsf{replicate} : (n : \mathbb{N}) \to A \to \mathsf{Vec}\ A\ n \\
&\mathsf{replicate}\ \mathsf{zero}\ \ \ x = [] \\
&\mathsf{replicate}\ (\mathsf{suc}\ n)\ x = x :: \mathsf{replicate}\ n\ x
\end{aligned}$$

E.g. $\mathsf{replicate}\ 3\ 0$ has type $\mathsf{Vec}\ \mathbb{N}\ 3$ and evaluates to $0 :: 0 :: 0 :: []$.

Dependent pattern matching

We can pattern match on Vec just like on List:

```
mapV : (A → B) → Vec A n → Vec B n
mapV f []       = []
mapV f (x :: xs) = f x :: mapV f xs

head : Vec A (suc n) → A
head (x :: xs) = x

tail : Vec A (suc n) → Vec A n
tail (x :: xs) = xs
```

In head and tail, the cases for [] are impossible!

A safe lookup

To define a total lookup on vectors, we need the type Fin n:

```
data Fin : ℕ → Set where
  zero : Fin (suc n)
  suc  : Fin n → Fin (suc n)

lookupVec : Vec A n → Fin n → A
lookupVec (x :: xs) zero    = x
lookupVec (x :: xs) (suc i) = lookupVec xs i
```

Again, there is no case for the empty vector!

The family of `Fin` types



Some more vector functions

Exercise. Implement the following functions:

$$\mathsf{zipVec} : \mathsf{Vec}\ A\ n \to \mathsf{Vec}\ B\ n \to \mathsf{Vec}\ (A \times B)\ n$$
$$\mathsf{updateVecAt} : \mathsf{Fin}\ n \to A \to \mathsf{Vec}\ A\ n \to \mathsf{Vec}\ A\ n$$

Outline

1 Introduction

2 Data types and (co)pattern matching

3 Universes and polymorphism

4 Dependent types

5 Coinductive record types

6 Mixing induction and coinduction

7 Sized types

Streams as a coinductive record

A stream consists of a head and a tail (= another stream):

```
record Stream (A : Set) : Set where
  coinductive
  field
    headS : A
    tailS : Stream A
open Stream public
```

The coinductive keyword indicates that we want the largest such type.

Functions on streams

We can use the projections to define functions on streams:

```
firstTwo : Stream A → A × A
firstTwo s = s .headS , s .tailS .headS

dropS : ℕ → Stream A → Stream A
dropS zero     s = s
dropS (suc n) s = dropS n (s .tailS)
```

Defining a new stream

Defining a new stream with a record constructor fails the termination check:

```
zeroes : Stream ℕ
zeroes = record { headS = o ; tailS = zeroes }
```

Termination checking failed for zeroes

Allowing this would violate strong normalization of Agda!

Defining a new stream

To define a new stream, we have to use copatterns instead:

```
zeroes : Stream ℕ
zeroes .headS = o
zeroes .tailS = zeroes
```

zeroes only reduces when projections are applied to it, thus preserving strong normalization.

The guardedness criterion

Coinductive values should be productive: applying any (finite) number of projections to them should terminate.

This is enforced by the guardedness criterion: [5] every (co)recursive call needs to appear

1. in a clause with a copattern, *and*
2. either at the top level or as the argument to one or more constructors

[5] Enabled by `{-# OPTIONS --guardedness #-}`

Limitations of guardedness

```
mapS : (A → B) → Stream A → Stream B
mapS f xs .headS = f (xs .headS)
mapS f xs .tailS = mapS f (xs .tailS)

nats : Stream ℕ
nats .headS = o
nats .tailS = mapS suc nats
```

Termination checking failed for nats

Problem. The guardedness checker does not know anything about mapS!

Working around the limitations

```
natsFrom : ℕ → Stream ℕ
natsFrom n .headS = n
natsFrom n .tailS = natsFrom (suc n)

nats : Stream ℕ
nats = natsFrom o
```

Alternatively, we can use sized types to provide Agda with more information about mapS (see later).

Exercises

- Define repeat : $A \to$ Stream A.
- Define lookup : Stream $A \to \mathbb{N} \to A$.
- Define tabulate : $(\mathbb{N} \to A) \to$ Stream A.
- Use tabulate to define fibonacci : Stream $\mathbb{N}$.
- Define transpose : Stream (Stream A) $\to$ Stream (Stream A).
- **(Bonus)** Prove that lookup (lookup (transpose xss) i) $j \equiv$ lookup (lookup xss j) i.

Outline

1 Introduction

2 Data types and (co)pattern matching

3 Universes and polymorphism

4 Dependent types

5 Coinductive record types

6 Mixing induction and coinduction

7 Sized types

Colists: potentially infinite lists

```
mutual
  data Colist (A : Set) : Set where
    []   : Colist A
    _::_ : A → Colist' A → Colist A

  record Colist' (A : Set) : Set where
    coinductive
    field
      force : Colist A
open Colist' public
```

Converting a stream to a colist

```
fromStream : Stream A → Colist A
fromStream {A} xs = xs .headS :: rest
  where
    rest : Colist' A
    rest .force = fromStream (xs .tailS)
```

Alternatively, we use a copattern lambda:

```
fromStream' : Stream A → Colist A
fromStream' {A} xs = xs .headS ::
  (λ where .force → fromStream' (xs .tailS))
```

Exercise. Define `fromList : List A → Colist A`.

Another example: co-natural numbers

```
mutual
  data CoℕN : Set where
    zero : CoℕN
    suc  : CoℕN' → CoℕN

  record CoℕN' : Set where
    coinductive
    field force : CoℕN
open CoℕN' public
```

Exercise. Define $\infty : \mathrm{Co}\mathbb{N}$, $\mathrm{from}\mathbb{N} : \mathbb{N} \to \mathrm{Co}\mathbb{N}$, and $\mathrm{colength} : \mathrm{Colist}\ A \to \mathrm{Co}\mathbb{N}$.

Outline

1 Introduction

2 Data types and (co)pattern matching

3 Universes and polymorphism

4 Dependent types

5 Coinductive record types

6 Mixing induction and coinduction

7 Sized types

Sized types

Sized types[6] are an alternative to the syntactic guardedness checker that annotates the size of expressions in their types.

The module `Size` provides:

- Size : Set
- Size< : Size → Set
- ∞ : Size.

+ some operators not relevant for coinduction.

[6] Enabled by `{-# OPTIONS --sized-types #-}`

Sized streams

We can parametrize a stream by its size i
(= the number of elements we can observe):

```
record Stream (A : Set) (i : Size) : Set where
  coinductive
  field
    headS : A
    tailS : {j : Size< i} → Stream A j
open Stream
```

The tail can be assigned any size j that is strictly smaller than i.

The infinite size

When *consuming* a stream we can use size ∞:

```
takeS : ℕ → Stream A ∞ → List A
takeS zero s     = []
takeS (suc n) s = s .headS :: takeS n (s .tailS)
```

Exercise. Define dropS.

Defining sized streams

When *defining* a new stream we should define it at an arbitrary size i:

```
zeroes : Stream ℕ i
zeroes .headS = o
zeroes .tailS = zeroes
```

More explicitly:

```
zeroes' : {i : Size} → Stream ℕ i
zeroes' {i} .headS = o
zeroes' {i} .tailS {j} = zeroes {j}
```

A size-preserving map

We can now give a more precise type to map:

```
mapS : (A → B) → Stream A i → Stream B i
mapS f s .headS = f (s .headS)
mapS f s .tailS = mapS f (s .tailS)
```

This allows a direct definition of nats:

```
nats : Stream ℕ i
nats .headS = o
nats .tailS = mapS suc nats
```

Exercise. Define zipWithS and use it to define fibonacci without using tabulate.

Next time

- The Curry-Howard correspondence
- Equational reasoning
- Bisimulation as a coinductive relation
- Bisimulation as the cubical path type